
In this section of the course we will look at some of the basic principles by
which metabolic networks operate and are controlled.

Control and regulation are two words that are frequently used in the lit-
erature in relation to cellular network studies. We don’t have an adequate
measure of what regulation is or in fact what it means. We do however
have a quantitative definition of what we mean by control.

In metabolism, control is the ability of an outside agent to influence a sys-
tem. The more influence an outside agent has the more control it has over
the network. Regulation is the mechanism that allows or does not allow the
influence to have control. To give two familiar examples. A driver of a car
has control over the speed and direction by way of an active regulatory sys-
tem that includes power steering and the automatic gear box. In contrast, a
cruise control allows the speed of a car to be stable even though the vehicle
may be forced to climb hills or speed up down hills. Sometimes regulation
is passive and other times active. Active regulation is more interesting be-
cause it will usually involve special structures or network motifs to enable
the regulation to occur.

1. A regulatory system that tracks is called a servo system.

2. A regulatory system that opposes control is called a homeostatic system.

Often systems are a combination of the two. For example, in order to
do tracking, some elements of the system are being maintained constant,
while in order to maintain hemostasis some elements of the system much
be changing.

How can be quantify these ideas?

In cellular networks it is customary to partition a subsystem, such as gly-
colysis or oxidative phosphorylation and study it in isolation. Under these
conditions it is possible to determine what the outside influences are and
what the internal regulating mechanisms are. Thus if we take a simple
metabolic pathway catalyzed by a series of enzymes, we can consider the
enzyme activities under the control of an outside agent (gene expression or
drug intervention), whereas the metabolite levels together with feedback



or feed-forward loops in the pathway are the regulating elements.

1. Computer Simulation Experiment: Simple First-Order Decay

We will be using Jarnac as the software tool to conduct metabolic simula-
tion experiments.

Screenshot of Jarnac: Top panel is the console; Bottom panel is the editor

Jarnac has two windows, an upper console and a lower editor window. All
models are entered into the lower window. Models can be controlled from
the upper window.

In the lower window enter the following model:

p = defn model
J1: S1 -> S2; k1*S1;

end;



p.S1 = 10;
p.S2 = 0;
p.k1 = 0.1;

m = p.sim.eval (0, 60, 100, [<p.time>, <p.S1>]);
graph (m);

To run this model click on the small green button in the center of the screen.

2. Simulating Two Steps

Let us now change the model and include an additional step:

p = defn model
J1: S1 -> S2; k1*S1;
J2: S2 -> S3; k2*S2;

end;

p.S1 = 10;
p.S2 = 0;
p.k1 = 0.1;
p.k2 = 0.2;

m = p.sim.eval (0, 60, 100, [<p.time>, <p.S1>, <p.S2>, <p.S3>]);
graph (m);

3. More realistic kinetics

In reality most kinetics are not irreversible. To make the last model more
realistic we will make each step reversible:

p = defn model
J1: S1 -> S2; k11*S1 - k12*S2;
J2: S2 -> S3; k21*S2 - k22*S3;

end;

p.S1 = 10;
p.S2 = 0;
p.k11 = 0.1; p.k12 = 0.08;
p.k21 = 0.2; p.k22 = 0.06;

m = p.sim.eval (0, 60, 100, [<p.time>, <p.S1>, <p.S2>, <p.S3>]);
graph (m);



What happens to the concentrations in this model?

What happens to the net reaction rates? We can output the reaction rates
by specifying the reaction name in the output list:

p = defn model
J1: S1 -> S2; k11*S1 - k12*S2;
J2: S2 -> S3; k21*S2 - k22*S3;

end;

p.S1 = 10;
p.S2 = 0;
p.k11 = 0.1; p.k12 = 0.08;
p.k21 = 0.2; p.k22 = 0.06;

m = p.sim.eval (0, 60, 100, [<p.time>, <p.S1>, <p.S2>, <p.S3>, <p.J1>, <p.J2>]);
graph (m);

What happens to the reaction rates?

4. Simulating an Open System

The previous examples illustrate models that are closed, that is no mass
enters or leaves the system. To simulate an open system we must fix the
boundaries. This is done easily by adding a $ symbol in front of each fixed
species.

In the model below the previous model was been modified by adding two
more reactions, one from a fixed source (Xo) and another to a fixed sink
(X1).

p = defn model
J1: $Xo -> S1; k0*Xo;
J2: S1 -> S2; k1*S1;
J3: S2 -> S3; k2*S2;
J4: S3 -> $X1; k3*S3;

end;

p.Xo = 10;
p.S1 = 0;
p.S2 = 0;
p.S3 = 0;
p.X1 = 0;

p.k0 = 0.05;



p.k1 = 0.1;
p.k2 = 0.2;
p.k3 = 0.34;

m = p.sim.eval (0, 60, 100, [<p.time>, <p.S1>, <p.S2>, <p.S3>, <p.J1>, <p.J2>]);
graph (m);

Explain what is happening to the metabolite concentrations and reaction
rates as the simulation proceeds?

5. Adding enzyme kinetics

We will now modify the model so that each reaction is governed by a re-
versible Michaelis-Menten rate law.

p = defn model
J1: $Xo -> S1; (Vm1/Km1)*(Xo - S1/Keq1)/(1 + Xo/Km1 + S1/Km2);
J2: S1 -> S2; (Vm2/Km3)*(S1 - S2/Keq2)/(1 + S1/Km3 + S2/Km4);
J3: S2 -> S3; (Vm3/Km4)*(S2 - S3/Keq3)/(1 + S2/Km5 + S3/Km6);
J4: S3 -> $X1; (Vm4/Km7)*(S3 - X1/Keq4)/(1 + S3/Km7 + X1/Km8);

end;

p.Xo = 10;
p.S1 = 0;
p.S2 = 0;
p.S3 = 0;
p.X1 = 0;

p.Vm1 = 1.5; p.Vm2 = 2.9; p.Vm3 = 10.5; p.Vm4 = 7.3;
p.Km1 = 1; p.Km2 = 1;
p.Km3 = 0.2; p.Km4 = 3.4;
p.Km5 = 0.9; p.Km6 = 1.2;
p.Km7 = 6.7; p.Km8 = 1.1;

p.Keq1 = 3.2;
p.Keq2 = 2.1;
p.Keq3 = 14.6;
p.Keq4 = 5.3;

m = p.sim.eval (0, 140, 200, [<p.time>, <p.S1>, <p.S2>, <p.S3>, <p.J1>, <p.J2>]);
graph (m);
p.ss.eval;

println p.J1;

There is a quick way to compute the steady state of a model:



p.ss.eval;

We can also print out variable values using the println command.

6. Perturbation experiments

What kinds of experiments can we do with the model in experiment 5? One
thing we can do is investigate how much each enzyme controls the level of
the steady state flux. To do this try the following:

1. Compute the steady state and record the steady state flux.
2. Increase the enzyme activity of E1 by 5%, i.e change Vm1 by 5%
3. Recompute the steady state flux and compute the % change in flux
4. Compute the ratio of J% to E%
5. Repeat for each enzyme step in the pathway.

Carry out this experiment and fill in the following table:

E% J% Ratio

Total:

In the last entry, total up the ratios and record the value.

The ratio J% to E% is called the flux control coefficient (CJ
E) and indicates

how much control a particular enzyme has over the flux. What determines
the individual values of CJ

E?

We can get some idea from the disequilibrium ratio. Given the following
reaction

S1 −→ S2

The equilibrium constant is given by:

Keq =
S2

S1



The mass-action ratio, Γ is given by:

Γ =
S2

S1

Γ is the ratio of metabolites at steady state. The disequilibrium ratio, ρ, is
then given by:

ρ =
Γ
Keq

If the step is near equilibrium then ρ ' 1, if the step is away from equilib-
rium then ρ� 1.

CJ
1 : CJ

2 : CJ
3 : CJ

4 = (1− ρ1) : ρ1(1− ρ2) : ρ1ρ2(1− ρ3) : ρ1ρ2ρ3(1− ρ4)

or for an arbitrary length pathway, the nth term is equal to:(
n−1∏
i=2

ρi

)
(1− ρn)

Consider the following question. If the ith reaction is irreversible what will
be its disequilibrium value? Hint: What would the value for the equilib-
rium constant be? Answer on next page.



The disequilibrium ratio for an irreversible step is ZERO (Keq is infinity, no
reactants at equilibrium). What effect will this have on the distribution of
control? For example, assume the second step is irreversible, ρ = 0. This
means that all terms from the third onwards are zero (ρ1ρ2(1 − ρ3). This
means that there is not control beyond the irreversible step? What is the
mechanistic explanation for this?

The second question is what happens to ρ is a reaction is at equilibrium?

Give an answer to the above, what effect does this have on the distribution
of control?

Finally, it is also possible to show that the disequilibrium ratio is equal to
the ratio of the reverse rate and forward rate:

ρ =
vr

vf

Show that the above relation is true for the simple reversible reaction:

S1
k1−⇀↽−
k2

S2

Since the forward rate will always to greater than the reverse rate for a
positive net rate, the disequilibrium ratio will always be less than one.

ρ ≤ 1

Since ρ is always less than one, the tendency is for most of the flux sensi-
tivity to be at the front of the pathway since downstream steps have more
and more multipliers of ρ values that are less than one.

Computing control coefficients by computer. A quick way to compute the
control coefficients is to use the commands:

println p.cc (<p.J1>, p.Vm1);
println p.cc (<p.J1>, p.Vm2);
println p.cc (<p.J1>, p.Vm3);
println p.cc (<p.J1>, p.Vm4);

7. Effect of negative feedback: Part I

Experiments have shown that only a fraction of control (20%) is found near
the front of a given pathway and most of the control is located at the end.
In this section we will investigate why this is so.



Let us change the previous model by adding a negative feedback loop to
the first step:

p = defn model
J1: $Xo -> S1; (Vm1/Km1)*(Xo - S1/Keq1)/(1 + Xo/Km1 + S1/Km2 + pow(S3,J0_h));
J2: S1 -> S2; (Vm2/Km3)*(S1 - S2/Keq2)/(1 + S1/Km3 + S2/Km4);
J3: S2 -> S3; (Vm3/Km4)*(S2 - S3/Keq3)/(1 + S2/Km5 + S3/Km6);
J4: S3 -> $X1; (Vm4/Km7)*(S3 - X1/Keq4)/(1 + S3/Km7 + X1/Km8);

end;

p.Xo = 10;
p.S1 = 0;
p.S2 = 0;
p.S3 = 0;
p.X1 = 0;
p.J0_h = 0; // This controls the strength of the feedback

p.Vm1 = 1.5; p.Vm2 = 2.9; p.Vm3 = 10.5; p.Vm4 = 7.3;
p.Km1 = 1; p.Km2 = 1;
p.Km3 = 0.2; p.Km4 = 3.4;
p.Km5 = 0.9; p.Km6 = 1.2;
p.Km7 = 6.7; p.Km8 = 1.1;

p.Keq1 = 3.2;
p.Keq2 = 2.1;
p.Keq3 = 14.6;
p.Keq4 = 5.3;

m = p.sim.eval (0, 140, 200,
[<p.time>, <p.S1>, <p.S2>, <p.S3>, <p.J1>, <p.J2>]);

graph (m);
p.ss.eval;
println p.cc (<p.J1>, p.Vm1);
println p.cc (<p.J1>, p.Vm2);
println p.cc (<p.J1>, p.Vm3);
println p.cc (<p.J1>, p.Vm4);

Your task is to investigate how the distribution of control changes as the
strength of the feedback is increased. Explain what you find..

8. Effect of negative feedback: Part II

p = defn feedback

J0: $X0 -> S1; J0_VM1*(X0-S1/J0_Keq1)/(1+X0+S1+pow(S4,J0_h));
J1: S1 -> S2; (10*S1-2*S2)/(1+S1+S2);
J2: S2 -> S3; (10*S2-2*S3)/(1+S2+S3);
J3: S3 -> S4; (10*S3-2*S4)/(1+S3+S4);
J4: S4 -> $X1; J4_V4*S4/(J4_KS4+S4);



end;

p.X0 = 10;
p.X1 = 0;
p.S1 = 0;
p.S2 = 0;
p.S3 = 0;
p.S4 = 0;
p.J0_VM1 = 10;
p.J0_Keq1 = 10;
p.J0_h = 1;
p.J4_V4 = 2.5;
p.J4_KS4 = 0.5;

m = p.sim.eval (0, 100, 200, [<p.time>, <p.S4>]);
graph (m);

Run this model and investigate its time dependent properties as you in-
crease the feedback strength.

1 Quantitative Theory

Useful reading material:

1. H Kacser, JA Burns, DA Fell - Biochemical Society Transactions, 1995 (Or
original from 1973) (Use Google scholar and search for ”control of flux”, it
will be the first match.

2. Anything written by David Fell (including his text book), Athel Cornish-
Bowden and Jannie Hofmeyr

3. A recent paper by Piero Morandini, on plant metabolism but a very fun
read. Rethinking metabolic control. Plant Science (2009), 176(4), 441-451

To make things simpler we will only consider properties at steady state.

Consider the simplest possible pathway:

Xo
v1−→ S

v2−→ X1

We will assume that the two boundary species, Xo and X1 are external and
fixed by the observer. In addition each reaction is catalyzed by an enzyme



at a rate v1 and v2 respectively by enzymes E1 and E2. The species S is
a state variable of the system and can change depending on the enzyme
activities, kinetic constants and boundary species concentrations. At steady
state the following relation will be true:

v1(Xo, S, E1, k1, . . .)− v2(S,X1, E2, k2, . . .) = 0

where the rates are expressed as functions of their influencing factors. There
is a simple graphical technique we can used to study how the enzyme ac-
tivities, E1 and E2 control the steady state concentration S, and the steady
state flux, J through the pathway. In this system system the steady state
flux, J will be numerically equal to the reaction rates v1 and v2,

J = v1 = v2

It is important to recall that for many enzyme catalyzed reactions the rate, v
is proportional to the concentration of enzyme, E, v ∝ E, if not, then there
is probably some other parameter that is.

Let us plot both reaction rates, v1 and v2 against the substrate concentration
S.

The above plot represents our reference state, in the plot we graph v1 and
v2 against S. The intersection point of the two curves marks the point when
v1 = v2, that is the steady state. A line dropped perpendicular to the inter-
section points marks the steady state concentration of S

Let us now change the activity of E2 by 30%. Note that the curve for v2
is scaled upwards, this in turns moves the intersection point to the left,
indicating that the steady state concentration of S decreases relative to the
reference state.



Let us now change the activity of E1 by 30%. Note that the curve for v1
is scaled upwards, this in turns moves the intersection point to the right,
indicating that the steady state concentration of S increases relative to the
reference state.

Let us now change the activity of both E1 and E2 by 30%. Note that the
curve for v1 and v2 are both scaled upwards, this in turns moves the inter-
section point upwards but doesn’t change the steady state concentration of
S. That is

δJ > 0 but δS = 0

This a very important result.



From these thought experiments we conclude that by increasing the activ-
ities of both enzymes by the same fraction we increase the flux through
the pathway but do not change the concentration of the pathway species,
S. This conclusion is in fact quite generate and no matter how complex
the pathway, if we increase the activity of every step in the pathway by a
given proportion, the same result will be obtained. Although we know that
δS = 0, how much has the flux increased under these conditions?

Since δS = 0, the only change that could possibly effect the flux is the
change in enzyme activity, since the enzyme activity has increased by a
given proportion (30%), then the flux must also have increased by the same
proportion since the rate is proportional to the enzyme activity. (i.e vi ∝
Ei).

Control Coefficients

Let us consider another thought experiment. Assume the pathway is at
steady state, let us now increase the activity of enzyme E1 by a factor α,
that is δE1/E1 = α. The system will now move to a new steady state with
a different flux and metabolite concentration. Let us assume that the flux
has changed by δJ and the species concentration by δS. We can measure
the influence of enzyme E1 on the flux and concentration of S by the ratio:

δJ

δE1

δS

δE1

However, these ratios depend on the size of the change and the units we
choose. We can eliminate these problems by taking infinitesimal changes
and scaling the ratio to remove units, when we do this we obtain what are
called the control coefficient:

CJ
Ei

=
dJ

dEi

Ei

J

C
Sj

Ei
=
dSj

dEi

Ei

Sj



Here we have generalized the definition for any enzyme step, Ei and flux,
J , and any species, Sj . The scaling values are obtained from the reference
state. The first coefficient is called the flux control coefficients and the sec-
ond one the concentration control coefficient. The values are dimensionless
and when measured give us some idea of how a particular enzyme controls
either the flux or some particular species. For a straight chain pathway one
can show that the flux control coefficients (CJ

E) are bounded between zero
and one.

There are various ways to look at the above definitions, these are given
below.

CJ
E =

d ln J
d lnE

CJ
E =

dJ

J
/
dE

E

CJ
E ≈

Percentage change in J
Percentage change in E

The last definition is a useful but approximate way to interpret the control
coefficients as a ratio of percentage changes.

The definitions given above can be rearranged and given the approxima-
tion:

δJ

J
= CJ

Ei

δEi

Ei

δSj

Sj
= C

Sj

Ei

δEi

Ei

These simple relations allow us to compute the change in flux given a
change in an enzyme activity. If we perturb more than one enzyme activity,
we can get the overall change by summing up the individual changes. In
general, if we make changes to n reaction steps, then the overall change in
flux and species concentrations is given by:



δJ

J
=

n∑
i=1

CJ
Ei

δEi

Ei

δS

S
=

n∑
i=1

CS
Ei

δEi

Ei

For the two step pathway, let us repeat the thought experiment where we
increased both enzyme activities at the same time. So long as we consider
small changes, we can compute the overall change in flux or species con-
centration by simply adding the control coefficient terms, thus:

δJ

J
= CJ

E1

δE1

E1
+ CJ

E2

δE2

E2

δS

S
= CS

E1

δE1

E1
+ CS

E2

δE2

E2

We can do this because the changes are assumed to be small and therefore
only the linear modes are stimulated.

However, we know from the thought experiments that δS = 0 and the
change in flux must equation the fractional change in enzyme activity, that
is δJ/J = δE1/E1 = δE2/E2 = α

Rewriting the above equations as:

α = CJ
E1
α+ CJ

E2
α

0 = CJ
E1
α+ CS

E2
α

from which we conclude:

1 = CJ
E1

+ CJ
E2

0 = CJ
E1

+ CS
E2



These summations (or theorems) are in fact general. For any pathway with
any topology the following is true:

Control Coefficient Summation Theorem

n∑
i=1

CJ
Ei

= 1

n∑
i=1

C
Sj

Ei
= 0

These theorems suggest the following:

1) Control is shared throughout a pathway.

2) If one step gains control, one of more other steps must loose control.

3) Control coefficients are system properties, they can only be computed or
measured in the intact system.

Rate-limiting Steps

In much of the literature and some contemporary textbooks, one will often
find a brief discussion of an idea called the rate-limiting step. The literature
is in general unclear about the meaning of this phrase but some interpret
the rate-limiting step to be the single step in pathway which limits the flux.
In terms of our control coefficients we can interpret the rate-limiting step
as the step with a flux control coefficient of unity. This means, by the sum-
mation theorem, that all other steps (at least in a linear chain) must have
flux control coefficients of zero. In reality no one has measured a control
coefficient of one, and in fact control is distributed throughout a pathway,
sometimes there may be ‘hot spots’ where more control is concentrated but
in general the notion of the rate-limiting step is a theoretical and experi-
mental fiction.

